



SOMMAIRE

Sommaire	1
News	2
Les derniers correctifs	2
Une macro pour lire plusieurs fichiers texte	3
Les sorties HTML avec le tagset TableEditor	7
Annexes	16

NEWS

LES DERNIERS CORRECTIFS

- [Liste des correctifs](#) mis en ligne entre le 21 mai et le 20 juin sur le Customer Support Center.
- [Le site des correctifs US](#)
- [Les correctifs spécifiques aux produits traduits en Français](#)

UNE MACRO POUR LIRE PLUSIEURS FICHIERS TEXTE

L'importation de plusieurs fichiers texte sous SAS peut être un traitement long. Ce traitement est d'autant plus long que les fichiers à importer sont nombreux et ne possèdent pas la même structure (nombre différent de variables, ordre des variables, etc..). SAS possède beaucoup d'outils puissants qui peuvent vous aider à automatiser ce traitement.

Caractéristiques :

Catégories : SAS/Base
OS : Windows, Unix
Version : toutes
Vérifié le 01/06/2007


Le but de cet article est de présenter une solution simple pour importer un ensemble de fichiers ne possédant pas forcément la même structure (nombre, ordre des variables différents).

La méthode présentée ci-dessous utilise uniquement le module SAS/BASE.

Exemple :


Nous souhaitons importer les deux fichiers présents dans le répertoire 'C:\Fichiers' au sein d'une table SAS nommée 'total.sas7bdat'.

Premier fichier que l'on souhaite importer :



Fichier	Edition	Format	Affichage	?	VAR1	VAR2	VAR3	VAR4	VAR5	VAR6	VAR7	VAR8	VAR9	VAR10	var11	var12	var13
111111	AAAA	1	1.1	10	5	1.1	2.1	3.1	4.1	5.1	6.1	7.1					
111111	BBBB	2	1.2	20	6	1.2	2.2	3.2	4.2	5.2	6.2	7.2					
111111	CCCC	3	1.3	30	7	1.3	2.3	3.3	4.3	5.3	6.3	7.3					
111111	DDDD	4	1.4	40	8	1.4	2.4	3.4	4.4	5.4	6.4	7.4					

Second fichier que l'on souhaite importer :



Fichier	Edition	Format	Affichage	?	VAR1	VAR2	VAR3	VAR4	VAR5	VAR7	VAR14	VAR12	VAR13
222222	AAAA	5	2.1	50	3.1	10	1	6.1					
222222	BBBB	6	2.2	60	3.2	20	1	6.2					
222222	CCCC	7	2.3	70	3.3	30	1	6.3					
222222	DDDD	8	2.4	80	3.4	31	1	6.4					

Le premier fichier contient 5 variables non présentes dans le second fichier à savoir : var6, var8, var9, var10 et var11.

Le second fichier contient lui une variable non présente dans le premier fichier à savoir var14. Ainsi seulement les cinq premières variables apparaissent dans le même ordre dans les deux fichiers.

On souhaite obtenir la table SAS suivante :

	VAR1	VAR2	VAR3	VAR4	VAR5	VAR6	VAR7	VAR8	VAR9	VAR10	VAR1	VAR12	VAR13	VAR14
1	111111	AAAA	1	1.1	10	5	1.1	2.1	3.1	4.1	5.1	6.1	7.1	
2	111111	BBBB	2	1.2	20	6	1.2	2.2	3.2	4.2	5.2	6.2	7.2	
3	111111	CCCC	3	1.3	30	7	1.3	2.3	3.3	4.3	5.3	6.3	7.3	
4	111111	DDDD	4	1.4	40	8	1.4	2.4	3.4	4.4	5.4	6.4	7.4	
5	222222	AAAA	5	2.1	50		3.1					1	6.1	10
6	222222	BBBB	6	2.2	60		3.2					1	6.2	20
7	222222	CCCC	7	2.3	70		3.3					1	6.3	30
8	222222	DDDD	8	2.4	80		3.4					1	6.4	31

Méthode :

1°) Récupération du nom des fichiers à importer

La première étape consiste à récupérer dans une table SAS le nom des fichiers à importer.

Pour cela l'instruction 'filename pipe' est utilisée.

Cette instruction permet d'exécuter une commande système et de récupérer son résultat dans une variable SAS.

On récupère également dans une macro variable le nombre de fichiers à importer.

Dans le programme ci-dessous, on récupère dans la table 'fichiers' tous les noms des fichiers présents dans le répertoire ' C:\Fichiers'.

On récupère également dans la macro variable 'nbr_fich' le nombre de fichiers à importer.

```
filename fich pipe 'dir C:\Fichiers /b';
```

```
data fichiers;
  length noms $20;
  infile fich truncover;
  input noms $20.;
  call symput ('nbr_fich',_n_);
run;
```

On obtient :

	noms
1	fichier1.txt
2	fichier2.txt

2°) Récupération du nom des variables

La macro lir_fich va ensuite lire chaque fichier successivement à l'aide d'une boucle %do %end sur le nombre de fichiers à importer.

Pour chaque fichier on récupère le nom des variables qu'il contient. Ceci correspond à la lecture de la première ligne du fichier.

Le nom des variables est alors stocké dans une table SAS sous la forme X1 à Xn, où n correspond au nombre maximum de variables présentes dans tous les fichiers.

Dans cette macro il est nécessaire de connaître le nombre maximum de variables présentes dans tous les fichiers que l'on souhaite importer.

Le programme 'nombre_max_variables.sas' en annexe permet de récupérer ce nombre dans une macro variable nommée nbr_var..

Dans cet exemple cette macro variable vaut 17.

```
%macro lir_fich;

  %do j=1 %to &nbr_fich;

    data _null_;
      set fichiers;
      if _n_=&j;
      call symput ('fichier',noms);
    run;

    %let nbr_var =17 ;

    data noms_var;
      length x1-x&nbr_var. $12;
      infile "C:\Fichiers\&fichier" obs=1 missover;
      input (x1-x&nbr_var. ) ($) ;
    run;
```

Lors de la lecture du premier fichier, le contenu de la table 'noms_var' est :

	x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	x13	x14	x15	x16	x17
1	VAR1	VAR2	VAR3	VAR4	VAR5	VAR6	VAR7	VAR8	VAR9	VAR10	VAR11	VAR12	VAR13				

Lors de la lecture du second fichier, le contenu de la table 'noms_var' est :

	x1	x2	x3	x4	x5	x6	x7	x8	x9	x10	x11	x12	x13	x14	x15	x16	x17
1	VAR1	VAR2	VAR3	VAR4	VAR5	VAR7	VAR14	VAR12	VAR13								

La macro noms_var est ensuite utilisée pour stocker le nom des variables dans des macros variables nommées V1 à V&nbr_var.

```
%macro noms_var;

  %do i=1 %to &nbr_var. ;
    %global v&i;

    data _null_;
      set noms_var;
      call symput("v&i",trim(x&i));
    run;
  %end;
%mend noms_var;

%noms_var;
```

3°) Lecture des données

La macro `lir_fich` lit ensuite les données de chaque fichier et les récupère dans une table temporaire nommé `temp`.

```
data temp;  
  infile " C:\Fichiers\&fichier" firstobs=2 missover;  
  input (&v1 &v2 &v3 &v4 &v5 &v6 &v7 &v8 &v9 &v10 &v11 &v12 &v13 &v14 &v15 &v16  
    &&v&nbr_var. ) ($);  
run;
```

Les fichiers sont ensuite regroupés dans une seule table nommée `total`.

```
%if &j=1 %then %do;  
  data total;  
    set temp;  
  run;  
%end;  
  
%else %do;  
  data total;  
    set total  
      temp;  
  run;  
  
%end;  
%end;  
  
%mend lir_fich;  
%lir_fich;
```

Pour plus d'informations, vous pouvez consulter le Technical Support Document 581, « Using FILEVAR= to read multiple external files in a DATA Step » (lien : <http://support.sas.com/techsup/technote/ts581.html>).

Auréliе GENTIL
Consultant Support Clients SAS France

LES SORTIES HTML AVEC LE TAGSET TABLEEDITOR

Le nouveau tagset TableEditor permet de créer des tableaux HTML attractifs. L'ODS offre ainsi de nouvelles fonctionnalités qui vont faciliter le travail tantôt du créateur de rapport (options plus simples d'utilisation que celles de la PROC TEMPLATE), tantôt l'utilisateur final (filtres, gel des en-têtes de colonnes et/ou de lignes, export vers Excel etc).



Caractéristiques :

Catégories : SAS/Base - ODS
OS : Windows
Version : SAS® 9.1.3
Vérifié le 20/06/2007

Cet article présente aux travers d'exemples les possibilités du tagset, les différentes options et l'installation du tagset.

Présentation

Le tagset TableEditor utilise l'activeX ScriptX pour personnaliser les pages HTML qu'il génère. Disponible uniquement avec Internet Explorer, il utilise les langages HTML et XML, les feuilles de styles et le JavaScript.

Aujourd'hui doté de plus de 90 options, il devrait encore s'enrichir prochainement. Les options portent soit sur les propriétés de la page, soit sur le tableau. Il n'y a pas d'option spécifique aux graphiques.

On distingue 7 familles d'options, que nous allons décliner dans les paragraphes suivants :

- Style de la page
- Export des données
- Options d'affichage
- Position et mise en forme du tableau
- Mise en page
- Table des matières
- Messages d'information

Dans les exemples donnés, seule l'instruction et les options ODS sont mentionnées. Le code complet est disponible en annexe 2.

Syntaxe

La syntaxe type :

```
ods tagsets.tableeditor file="CheminEtNom.html" style=science
options(option1='val' option2='val');
    ...instructions générant une sortie : procédures PRINT, TABULATE, FREQ
    etc ...
ods tagsets.tableeditor close;
```

Si le chemin du fichier n'est pas spécifié (FILE), alors il sera créé sous C:\Documents and Settings_identifiant_.

Si le nom du fichier n'est pas spécifié, alors il sera généré sous la forme sashtml.htm (puis sashtml2.htm, sashtml3.htm ... pour les exécutions suivantes).

Si le style n'est pas spécifié, alors le style sasuser.default sera utilisé.

Les options de style de la page et du tableau

Les options

- Choix des couleurs pour la trame de fond : il est possible d'alterner la couleur une ligne sur deux (options BANNER_COLOR_EVEN et BANNER_COLOR_ODD) et/ou une colonne sur deux (COL_COLOR_EVEN et COL_COLOR_ODD). Cette fonctionnalité est également réalisable via la procédure TEMPLATE, cf . [FAQ 3597 \(lien : http://support.sas.com/faq/035/FAQ03597.html\)](http://support.sas.com/faq/035/FAQ03597.html).
- Grille du tableau : choix de la couleur (GRIDLINE_COLOR), ou choix de ne pas afficher de grille (GRIDLINE)

- Ajout et positionnement d'une image au-dessus du tableau (IMAGE_PATH et IMAGE_JUST)
- Choix de la police de caractère de toute la page (FONTFAMILY)

Illustration

Ce premier exemple montre l'alternance des couleurs des colonnes pour en faciliter la lecture :

Résultat des ventes par pays						
	CANADA		GERMANY		U.S.A.	
	Prévision	Réalisé	Prévision	Réalisé	Prévision	Réalisé
BED	471.67	455.08	480.00	489.96	525.90	502.10
CHAIR	500.17	509.63	478.98	482.19	468.73	529.44
DESK	487.67	534.90	412.94	524.13	533.29	527.79
SOFA	510.73	556.67	563.33	640.17	519.23	458.00
TABLE	515.79	464.98	504.73	517.81	531.25	504.60

Pour la page HTML ci-dessus, les options utilisées sont :

- la couleur de la trame de fond à afficher une colonne sur deux
- la couleur du fond de la page
- la police

```
ods tagsets.tableeditor
file="col_color_even.html"
options ( col_color_even="99FF66"
col_color_odd="99CCFF"
background_color="FFFFCC"
fontfamily="Arial" );
```

Ce deuxième exemple montre l'alternance des couleurs des lignes :

<i>Résultat des ventes par année</i>						
Year=1993						
	CANADA		GERMANY		U.S.A.	
	Prévision	Réalisé	Prévision	Réalisé	Prévision	Réalisé
BED	471.67	455.08	480.00	489.96	525.90	502.10
CHAIR	500.17	509.63	478.98	482.19	468.73	529.44
DESK	487.67	534.90	412.94	524.13	533.29	527.79
SOFA	510.73	556.67	563.33	640.17	519.23	458.00
TABLE	515.79	464.98	504.73	517.81	531.25	504.60

<i>Résultat des ventes par année</i>						
Year=1994						
	CANADA		GERMANY		U.S.A.	
	Prévision	Réalisé	Prévision	Réalisé	Prévision	Réalisé
BED	449.48	539.27	432.42	471.17	512.77	501.52
CHAIR	474.75	537.02	439.13	499.17	473.88	531.73
DESK	541.35	552.33	517.04	486.33	553.44	483.52
SOFA	441.90	487.81	468.27	506.92	422.60	446.02
TABLE	461.06	507.94	527.21	507.13	494.79	460.04

Pour la page HTML ci-dessus, les options utilisées sont :

- la couleur de la trame de fond à afficher une ligne sur deux
- la couleur du fond de la page
- la police

```
ods tagsets.tableeditor
file="!temp\banner_color.html"
options ( banner_color_even="FFCCFF"
banner_color_odd="CC99FF"
background_color="FFFFCC"
fontfamily="Arial" );
```

L'export des données

Les options

- Personnalisations possibles pour les exports vers Excel : gel des colonnes (EXCEL_FROZEN_HEADERS), nom des feuilles (SHEET_NAME), boîte de dialogue (EXCEL_SHEET_PROMPT, EXCEL_SAVE_PROMPT, EXCEL_SAVE_DIALOG, EXCEL_SAVE_FILE), orientation (EXCEL_ORIENTATION), filtres (EXCEL_AUTOFILTERS), zoom (EXCEL_SCALE, EXCEL_ZOOM), création ou mise à jour (UPDATE_TARGET) ...
- Format d'export (FILE_FORMAT) : .xls, .txt, .csv, .doc, .xml, .slk
- Gestion des pivots

Illustration

Lorsque l'on utilise des options d'export, un bouton « Export » est ajouté à la page HTML :

Export

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0

Une fois que l'on a cliqué sur ce bouton, est que l'on a éventuellement répondu aux boîtes de dialogue pour sauvegarder le fichier Excel, alors la feuille Excel se présente comme ceci :

	A	B	C	D	E	F
1						
2						
3	Obs	Name	Sex	Age	Height	Weight
4	1	Alfred	M	14	69.0	112.5
5	2	Alice	F	13	56.5	84.0
6	3	Barbara	F	13	65.3	98.0
7	4	Carol	F	14	62.8	102.5

Pour la page HTML ci-dessus, les options utilisées sont :

- l'ajout du bouton « export » sur la page
- ajout des filtres dans la feuille Excel
- gel des colonnes dans la feuille Excel
- nom de la feuille

```
ods tagsets.tableeditor
file="export.html"
style=styles.analysis
options ( excel_save_dialog="yes"
excel_autofilter="yes"
excel_frozen_header="yes"
sheet_name="class" ) ;
```


Les options d'affichage

Les options

- Autoriser le glisser/lâcher (DRAG) permet de copier le texte sélectionné sur la page et le coller dans un fichier texte simplement par la souris
- Zoom sur l'ensemble de la page (ZOOM) ou sur un choix de tableaux (TABLE_ZOOM)
- Tri des colonnes en cliquant sur l'intitulé (SORT). Personnalisation des colonnes pouvant être triées (SORT_ARROW, SORT_IMAGE, HEADER_UNDERLINE)
- Ajout d'une couleur sur l'intitulé des colonnes (DESCRIBE) pour indiquer leur type (réalisable via la procédure TEMPATE, cf. FAQ 3223 (lien : <http://support.sas.com/fag/032/FAQ03223.html>)).
- Autoriser l'édition de la page (DESIGN_MODE)
- Ajout de filtres (AUTOFILTER) sur les colonnes voulues (FILTER_COLS)
- Choix de la feuille de style à appliquer (STYLE_SWITCH)

Illustration

Dans cet exemple, le type des colonnes est mis en avant, les colonnes peuvent être triées en cliquant sur l'intitulé (ici le tableau est trié par l'âge), et un menu a été ajouté au sein même de la page pour effectuer des actions :



Obs	Name	Sex	Age ▲	Height	Weight
11	Joyce	F	11	51.3	50.5
18	Thomas	M	11	57.5	85.0
6	James	M	12	57.3	83.0
7	Jane	F	12	59.8	84.5

Pour la page HTML ci-dessus, les options utilisées sont :

- le glisser-lâcher
- le tri des colonnes
- ajout d'un signe sur les en-têtes de colonnes pour donner le type de la variable
- ajout d'un menu pour sauver ou imprimer le fichier

```
ods tagsets.tableeditor  
file="Sort.html" style=styles.analysis  
options ( drag="yes"  
sort="yes"  
describe="yes"  
print_dialog="yes" ) ;
```

Dans ce second exemple, un filtre est ajouté aux 2 premières colonnes. Ici, on visualise les données des jeunes filles :

Elèves de la classe

Obs	Name	Sex	Age	Height	Weight
▼	(all) ▼	▼	▼	▼	▼
2	Alice	(all) f m	13	56.5	84.0
3	Barbara	F	13	65.3	98.0
4	Carol	F	14	62.8	102.5
7	Jane	F	12	59.8	84.5
8	Janet	F	15	62.5	112.5
11	Joyce	F	11	51.3	50.5
12	Judy	F	14	64.3	90.0
13	Louise	F	12	56.3	77.0
14	Mary	F	15	66.5	112.0

Pour la page HTML ci-dessus, les options utilisées sont :

- l'ajout de filtres
- sélection des colonnes où le filtre est actif.

```
ods tagsets.tableeditor
file="!temp\Filtres.html"
style=styles.statdoc
options(autofilter="yes"
filter_cols="1,2" ) ;
```

A noter : les options de filtres fonctionnent mal avec le tagset de mars 2007. Elles sont corrigées dans la prochaine version du tagset.

Les options de position et de mise en forme du tableau

Les options

- Taille (PAGEHEIGHT, PAGEWIDTH) et couleur de l'ascenseur (SCROLLBAR_COLOR) : en spécifiant une taille, le tableau apparaît dans un encadré doté d'un ascenseur de la taille choisie.
- Gel des en-têtes de colonnes (FROZEN_HEADERS) et/ou des lignes (FROZEN_ROWHEADERS) : lors de l'impression, les en-têtes de colonnes et/ou de lignes seront imprimées sur chaque page

Illustration

Dans cette page, le tableau est dans un encadré ayant son propre ascenseur. L'en-tête et la grille du tableau sont en vert. L'impression se fera en paysage, et les en-têtes de colonnes seront répétés sur chaque page.

Description des tables de la SASHELP

memname	crdate	modate	nlobs	obslen	nvar	indxtype	audit
HTHEME	12MAY04	12MAY04	3	22	2	SIMPLE	no
HTMUTIL	13MAY04	13MAY04	6	128	3		no
IMAGERW	12MAY04	12MAY04	21	216	13		no
IMGMSG	12MAY04	12MAY04	112	288	6	SIMPLE	no
JAC	12MAY04	12MAY04	7923	22	5		no
KOC	12MAY04	12MAY04	2561	18	4		no
LANKB	15FEB06	15FEB06	147	103	6		no
LIBRARY	12MAY04	12MAY04	3	888	18		no
LSWMSG	12MAY04	12MAY04	49	288	6	SIMPLE	no
LTHEME	12MAY04	12MAY04	13	22	2	SIMPLE	no
MACRS10	12MAY04	12MAY04	11	16	2		no
MACRS15	12MAY04	12MAY04	16	16	2		no
MACRS20	12MAY04	12MAY04	21	16	2		no

L'aperçu avant impression (les en-têtes de colonnes sont bien répétées sur chaque page) :

memname	crdate	modate	nlobs	obslen	nvar	indxtype	audit
EISOBJ1	12MAY04	12MAY04	10	312	13	SIMPLE	no
EISOBJ2	12MAY04	12MAY04	4	312	13	SIMPLE	no
EISOBJ3	12MAY04	12MAY04	5	312	13	SIMPLE	no
EISOBJ4	12MAY04	12MAY04	4	312	13	SIMPLE	no
EISOBJ5	12MAY04	12MAY04	8	312	13	SIMPLE	no
EISOBJ6	12MAY04	12MAY04	1	312	13	SIMPLE	no
EISOBJ7	12MAY04	12MAY04	10	312	13	SIMPLE	no

A noter : comme pour l'ODS HTML classique, la trame de fond n'est pas imprimée. De plus, le rendu dépend de la configuration de l'imprimante.

Pour la page HTML ci-dessus, les options utilisées sont :

- la couleur du fond de la page
- la police
- la trame de fond des en-têtes de colonne
- la couleur de la grille du tableau
- le gel des en-têtes de colonnes
- le mode paysage à l'impression
- la hauteur du tableau avec ascenseur

```
ods tagsets.tableeditor
file="!temp\frozen_headers.html"
options (background_color="white"
fontfamily="Arial"
header_bcolor="66CC99"
gridline_color="66CC99"
frozen_headers="yes"
orientation="landscape"
pageheight="12 cm" );
```

Les options de mise en page

Les options

Les options de mise en page permettent de personnaliser l'en-tête (PRINT_HEADER), le pied-de-page (PRINT_FOOTER), l'orientation de la page à l'impression (ORIENTATION), les marges (LEFT_MARGIN, RIGHT_MARGIN, TOP_MARGIN, BOTTOM_MARGIN), la réduction ou l'agrandissement à l'impression (PRINT_ZOOM, FIT2PAGE)

Illustration

La page HTML a été créée avec une en-tête et un pied de page, en mode portrait. Un zoom permet de s'assurer que le tableau ne sera pas tronqué en largeur à l'impression. L'aperçu avant impression (*image scindée en 2*) :

memname	crdate	modate	nlobs	obslen	nvar	indxtyp	audit
ADGPHG	12FAY08	12FAY08	428	260	8	SIMPLE	NO
ADPHG	12FAY08	12FAY08	428	260	8	SIMPLE	NO
APPHG	12FAY08	12FAY08	1088	288	8	SIMPLE	NO
ACS	12FAY08	12FAY08	184	18	2		NO
ASRCPRR	12FAY08	12FAY08	802	892	18		NO
ASRCMA	12FAY08	12FAY08	17388	188	18	COMPOSITE	NO
ESYDGLS	12FAY08	12FAY08	8	260	18	COMPOSITE	NO
ETLRHS	27FAY08	27FAY08	28	128	8		NO
ETPHG	12FAY08	12FAY08	8	260	8	SIMPLE	NO

Pour la page HTML ci-dessus, les options utilisées sont :

- l'en-tête de la page
- le pied-de-page
- le mode portrait pour l'impression
- un zoom sur le tableau

```
ods tagsets.tableeditor  
file="Page_Setup.html" style=science  
options (Print_Header='Généré le &d'  
Print_Footer='&b&b Page &p sur &P'  
orientation="portrait"  
print_zoom="0.75" );
```

La table des matières

La table des matières peut être personnalisée avec l'ajout d'images (OPEN_IMAGE_PATH, CLOSED_IMAGE_PATH, LEAF_IMAGE_PATH), le choix de la trame de fond (TOC_BACKGROUND), ou encore l'ajout d'un bouton « Impression » pour le corps de la page (TOC_PRINT)

Les messages d'information

D'autres options permettent :

- o L'ajout de fichier dans le corps de la page de type HTML, RTF ou PDF (INCLUDE)
- o Le choix du titre de la fenêtre (WINDOW_TITLE)
- o L'affichage d'un message d'attente lors du chargement de la page (LOAD_MSG),
- o L'affichage d'un message d'avertissement une fois la table chargée (ALERT_TEXT).

Pré-requis et installation

La personne qui crée la page HTML doit utiliser SAS@9. Pour installer le tagset, les étapes à suivre sont :

- Récupérer le tagset (lien : <http://support.sas.com/rnd/base/topics/odsmarkup>) : télécharger le fichier compressé contenant le fichier tableeditor.pl
- Ouvrez le tableeditor.pl depuis la fenêtre Enhanced Editor depuis SAS et soumettre le code (PROC TEMPLATE).

La personne qui consulte la page n'a pas d'installation à réaliser, mais doit respecter les pré-requis suivants :

- Utiliser Internet Explorer
- Autoriser l'exécution d'ActiveX :
 - o IE > Outils > Options > Sécurité > Personnaliser le niveau : cocher « Exécuter les contrôles ActiveX et les plug-ins »
 - o Sous Windows XP, lorsqu'une barre d'avertissement indique « *Pour vous aider à protéger votre ordinateur, Internet Explorer a restreint l'affichage du contenu actif de ce fichier, qui pourrait accéder à votre ordinateur.* », il faut cliquer sur la barre et choisir « Autoriser le contenu bloqué », puis valider le choix.

Liens utiles

Présentation, installation et documentation complète du tagset TableEditor (lien : <http://support.sas.com/rnd/base/topics/odsmarkup/tableeditor/index.html>)

Liste des couleurs : TS-688 « Defining Colors using Hex Values » (lien : <http://support.sas.com/techsup/technote/ts688/ts688.html>)

L'activeX ScriptX (lien : <http://www.meadroid.com/scriptx/>)

L'activeX TableEditor devrait prochainement s'enrichir de nouvelles fonctionnalités, telles que la génération de diapositives Powerpoint ou encore le masquage de colonne. Les filtres automatiques seront améliorés. N'hésitez pas à consulter le site <http://support.sas.com/rnd/base/topics/odsmarkup> pour vous tenir informé !

Géraldine Deschamps
Consultant Support Clients
SAS France

ANNEXES

Annexe 1 : programme nombre variable max.sas

```
/* Exemple de programme permettant de récupérer dans une macro variable  
nommée nbr_var le nombre maximum de variables  
présentés dans les 2 fichiers (nommés fichier1.txt et fichier2.txt dans  
notre exemple) que l'on souhaite importer */
```

```
data tab1 ;  
  infile "C:\fichiers\fichier1.txt" obs=1 missover lrecl=72;  
  length var $72;  
  input @1 VAR $ 1-72 ;  
  VAR =trim(left(compbl(VAR )));  
run;
```

```
data tab2 ;  
  infile "C:\fichiers\fichier2.txt" obs=1 missover lrecl=49;  
  length var $72;  
  input @1 VAR $ 1-49 ;  
  VAR =trim(left(compbl(VAR )));  
run;
```

```
data var1 (drop=i);  
  set tab1;  
  retain nbvar1 0;  
  do i=1 to length(var);  
    if substr(var ,i,1)=" " then nbvar1=nbvar1+1;  
  end;  
run;
```

```
data var2 (drop=i);  
  set tab2;  
  retain nbvar2 0;  
  do i=1 to length(var);  
    if substr(var,i,1)=" " then nbvar2=nbvar2+1;  
  end;  
run;
```

```
data var1;  
  set var1;  
  length vart1 $10;  
  do i=1 to nbvar1+1;  
    vart1=scan(var ,i);output;  
  end;  
run;
```

```
data var2;  
  set var2;  
  length vart2 $10;  
  do i=1 to nbvar2+1;  
    vart2=scan(var,i);output;  
  end;  
run;
```

```
data final ;  
  merge var1 (keep=vart1 i)  
        var2 (keep=vart2 i);  
  by i;
```

```

run;

data final2 ;
  set final;
  retain nbr_var 0;
  if vart1 eq vart2 then nbr_var=nbr_var+1;
  else if vart1 ne vart2 and vart2 ne '' and vart1 ne '' then
nbr_var=nbr_var+2;
  else nbr_var=nbr_var+1;
  call symput('nbr_var',put(nbr_var,8.));
run;

%put nbr_var=&nbr_var;

```

Annexe 2 : programmes complets de l'article sur le tagset TableEditor

```

/* Exemple 1 : les options de style */

ods listing close ;
ods tagsets.tableeditor file="!temp\col_color_even.html"
options (col_color_even="99FF66"
         col_color_odd="99CCFF"
         background_color="FFFFCC"
         fontfamily="Arial"
        );

proc tabulate data=sashelp.prdsale (where=(year=1993));
  class country region product;
  var actual predict;
  tables product=' ' ,country=' '* (predict="Prévision"
actual="Réalisé")*mean=' ' ;
  title "Résultat des ventes par pays" ;
run;

ods tagsets.tableeditor close;

ods tagsets.tableeditor file="!temp\banner_color.html"
  options (banner_color_even="FFCCFF"
         banner_color_odd="CC99FF"
         background_color="FFFFCC"
         fontfamily="Arial"
        );

title "Résultat des ventes par année" ;

proc sort data=sashelp.prdsale out=prdsale ;
  by year country region product ;
run ;

proc tabulate data=prdsale ;
  class country region product;
  var actual predict;
  by year ;
  tables product=' ' ,country=' '* (predict="Prévision"
actual="Réalisé")*mean=' ' ;
run;

```

```

ods tagsets.tableeditor close;

/* Exemple 2 : Export */

ods tagsets.tableeditor file="!temp\export.html" style=styles.analysis
  options(excel_save_dialog="yes"
    excel_autofilter="yes"
    excel_frozen_header="yes"
    sheet_name="class"
  ) ;
title " " ;
proc print data=sashelp.class ;
run ;

ods tagsets.tableeditor close;

/* Exemple 3 : options d'affichage */

ods tagsets.tableeditor file="!temp\Sort.html" style=styles.analysis
  options(drag="yes"
    sort="yes"
    describe="yes"
    print_dialog="yes"
  ) ;
proc print data=sashelp.class ;
run ;

ods tagsets.tableeditor close;

ods tagsets.tableeditor file="!temp\Filtres.html" style=styles.statdoc
  options(autofilter="yes" filter_cols="1,2"
  ) ;
title "Elèves de la classe" ;
proc print data=sashelp.class ;
run ;

ods tagsets.tableeditor close;

/* Exemple 4 : les options de position et mise en forme */

ods tagsets.tableeditor file="!temp\frozen_headers.html"
  options(header_bcolor="66CC99"
    gridline_color="66CC99"
    background_color="white"
    fontfamily="Arial"
    frozen_headers="yes"
    orientation="landscape"
    pageheight="12 cm"
  );

title "Description des tables de la SASHELP" ;
proc print data=sashelp.vtable noobs ;
  where libname='SASHELP' and memtype='DATA' ;

```

```

var memname crdate modate nlobs obslen nvar indxtype audit ;
format crdate modate datetime8. ;
run ;

ods tagsets.tableeditor close;

/* Exemple 5 : les options de mise en page */
ods tagsets.tableeditor file="!temp\Page_Setup.html" style=science
options(Print_Header='Généré le &d'
Print_Footer='&b&b Page &p sur &P'
orientation="portrait"
print_zoom="0.75"
);
title "Description des tables de la SASHELP" ;
proc print data=sashelp.vtable noobs ;
where libname='SASHELP' and memtype='DATA' ;
var memname crdate modate nlobs obslen nvar indxtype audit ;
format crdate modate datetime8. ;
run ;

ods tagsets.tableeditor close;

```

Directeurs de la publication :

Philippe Hoffmann
Géraldine Deschamps

Comité de rédaction :

Aurélie Gentil
Géraldine Deschamps

Comité de relecture

Fabienne Bernard
Mouloud Dey
Philippe Hoffmann



THE
POWER
TO KNOW.

SAS FRANCE - DOMAINE DE GRÉGY - BP 5 - 77166 GRÉGY-SUR-YERRES - FRANCE

TÉL. : +33 (0) 1 60 62 11 11 - FAX : +33 (0) 1 60 62 11 99 WWW.SAS.COM/FRANCE

SAS et tous les autres noms de produits ou de services SAS Institute Inc. sont des marques ou des marques déposées de SAS Institute Inc. aux États-Unis et dans d'autres pays. ® indique que la marque est déposée aux États-Unis. Les autres appellations et noms de produits sont des marques de leurs sociétés respectives. Copyright © 2006, SAS Institute Inc. Tous droits réservés.